

Tema 2

Càlculs bàsics amb wxMaxima

Objectius

1. Dur a terme càlculs bàsics amb variables.
2. Dur a terme càlculs bàsics amb expressions algebraiques.
3. Dur a terme càlculs bàsics amb polinomis i fraccions racionals.
4. Resolució bàsica d'equacions i inequacions.
5. Funcions elementals: definició i sintaxi amb wxMaxima.

Continguts

- 02-1. Càlculs bàsics amb variables.
- 02-2. Expressions algebraiques.
- 02-3. Polinomis i fraccions racionals
- 02-4. Resolució bàsica d'equacions.
- 02-5. Resolució bàsica de sistemes d'equacions.
- 02-6. Resolució bàsica d'inequacions.
- 02-7. Funcions elementals amb wxMaxima.

Referències

- AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P.
(2011)
Cálculo con wxMaxima.
- G07 GLASNER, MOSES (2007)
A Maxima Guide for Calculus Students
- JB01 JARAUTA BRAGULAT, EUSEBI (2001)
Anàlisi Matemàtica d'una variable. Fonaments i aplicacions.
- RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J. RAFAEL
(2005)
Introducción a Maxima
- RG07 RODRÍGUEZ GALVÁN, J. RAFAEL (2007)
Maxima con wxMaxima: software libre en el aula de matemáticas

- RR08 RODRÍGUEZ RIOTORTO, MARIO (2008)
Primeros pasos en Maxima
- RV09 RODRÍGUEZ, MARIO; VILLATE, JAIME (2009)
Manual de Maxima.
- VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

02-1.- Càlculs bàsics amb variables

Els càlculs d'aquest apartat es desenvolupen al fitxer **Tema_02-1.wxm**.

Sovint els càlculs matemàtics s'estructuren i es duen a terme amb variables, és a dir, símbols que poden prendre valors numèrics diferents. Per designar una variable es pot fer servir una lletra, un conjunt de lletres o un conjunt de lletra i número. Són exemples de variables: x, y, z, xx, yy, x1, x2, y1, y2, etc.

La sintaxi de l'assignació de variables amb wxMaxima és senzilla: es tracta d'escriure el símbol o símbols que es vol aplicar a la variable, a continuació el signe ":" i finalment el valor numèric assignat. Si es posa al final el signe "\$" s'assignarà el valor a la variable però no es mostrarà i si al final es posa el signe ";" s'assignarà i es mostrarà. Vegem uns exemples de variables i d'assignació de valors numèrics a variables, amb realització d'uns càlculs elementals:

```
(%i1) a1:115.654$      b1:401.928$
(%i3) a:234.34567;      b:567.4321;
(%o3) 234.34567
(%o4) 567.4321
(%i5) a+b;      a*b;      2.145*a-34.45*b;
(%o5) 801.77777
(%o6) 132975.255654007
(%o7) -19045.36438285
```

Sintaxi per mostrar el símbol d'una variable a més del seu valor numèric, assignat o calculat:

```
(%i8) '(a+b)=a+b;      '(ab)=a*b;
      '(2.145*a-34.45*b)=2.145*a-34.45*b;
(%o8) b + a = 801.77777
(%o9) ab = 132975.255654007
(%o10) 2.145 a - 34.45 b = -19045.36438285
```

Exemple de com es poden considerar o definir noves variables, assignar valors numèrics a aquestes variables i calcular el valor d'una expressió en la qual intervenen aquestes variables noves i variables anteriors:

```
(%i11) x:987.65$      y:432.19$
      45.34*a+24.345*b-67.765*x^2+34.21*sqrt(y);
(%o13) -6.6076389622216567 107
```

Il·lustració de com es poden definir variables noves a partir d'altres variables definides amb anterioritat i com es poden fer càlculs amb aquestes variables:

```

(%i14) R1:x^2+y^2;          R2:x+y;          R1+R2;
(%o14) 1162240.7186
      (%o15) 1419.84
      (%o16) 1163660.5586
(%i17) '(R1)=R1:x^2+y^2;    '(R2)=R2:x+y;
      '(R1+R2)=R1+R2;
(%o17) R1 = 1162240.7186
      (%o18) R2 = 1419.84
      (%o19) R2 + R1 = 1163660.5586

```

Exemple de l'aplicació de les variables a un procediment per al càlcul de l'àrea d'un cercle:

```

(%i20) Radi:23.456;        '(Area_cercl)=%pi*Radi^2, numer;
(%o20) 23.456
      (%o21) Area_cercl = 1728.453811460717

```

Exemple de l'aplicació de les variables a un procediment per al càlcul de l'àrea d'un triangle:

```

(%i22) Base:123.45;        Altura:6.7;
      '(Area_triangle)=Base*Altura/2;
(%o22) 123.45
      (%o23) 6.7
      (%o24) Area_triangle = 413.5575

```

Exemple de l'aplicació de les variables a un procediment per al càlcul de l'àrea d'un trapezi:

```

(%i25) B1:12.34;           B2:7.543;
      H:2.345;              '(Area_Trapezi)=(B1+B2)*H/2;
(%o25) 12.34
      (%o26) 7.543
      (%o27) 2.345
      (%o28) Area_Trapezi = 23.3128175

```

Pot interessar modificar el valor assignat a una variable o a diverses variables. Amb aquesta finalitat cal aplicar la instrucció “kill(all)\$” per anul·lar (“matar”) els valors assignats o calculats de les variables definides o considerades.

A partir de l'execució d'aquesta instrucció les variables no tenen cap valor numèric assignat fins que es faci una nova assignació de manera específica. Si s'elaboren diversos càlculs amb una certa variable, es recomana anul·lar el seu valor abans d'iniciar un nou

procediment. Així, per exemple en relació a les variables aplicades en càlculs anteriors es tindria:

```
(%i29) kill(all)$  
      a+b;      a*b;  
      R1;      R2;  
      Radi;     Base;  
(%o1) b + a  
(%o2) a b  
(%o3) R1  
(%o4) R2  
(%o5) Radi  
(%o6) Base
```

02-2.- Expressions algebraiques

En les sessions anteriors s'ha vist com es duen a terme càlculs aritmètics elementals i càlculs amb variables. Es tracta ara de veure com el programa es capaç de treballar amb expressions algebraiques.

Els càlculs d'aquest apartat es desenvolupen al fitxer **Tema_02-2.wxm**.

Exemple d'expressió algebraica:

```
(%i1) a^2 + b/a + b;  
(%o1)  $\frac{b}{a} + b + a^2$ 
```

Quan es vol aplicar una instrucció al resultat immediatament anterior es pot fer mitjançant la sintaxi instrucció(%), on “instrucció” indica la funció a realitzar. Així, per exemple, si es vol simplificar el resultat anterior i es vol fer amb la instrucció “ratsimp” (simplificar):

```
(%i2) ratsimp(%);  
(%o2)  $\frac{(a+1)b+a^3}{a}$ 
```

De la mateixa manera, es pot expandir una fracció algebraica, que no és ni més ni menys que l'invers del procediment anterior:

```
(%i3) expand(%);  
(%o3)  $\frac{b}{a} + b + a^2$ 
```

Exemples de potències de binomis:

```
(%i4) (a+b)^2;  
(%o4)  $(b+a)^2$   
(%i5) expand(%);  
(%o5)  $b^2 + 2ab + a^2$   
(%i6) (a+b)^4;  
(%o6)  $(b+a)^4$   
(%i7) expand(%);  
(%o7)  $b^4 + 4ab^3 + 6a^2b^2 + 4a^3b + a^4$   
(%i8) (a+b)^6;  
(%o8)  $(b+a)^6$ 
```

```
(%i9) expand(%);
(%o9) b6+6 a b5+15 a2 b4+20 a3 b3+15 a4 b2+6 a5 b+a6
```

Exemples d'operacions amb expressions algebraiques:

```
(%i10) (a+b)^6 -8*(a+b)^4 + 10;
(%o10) (b+a)6-8(b+a)4+10
(%i11) expand(%);
(%o11) b6+6 a b5+15 a2 b4-8 b4+20 a3 b3-32 a b3+15 a4 b2-48 a2 b2+6 a5 b-32 a3 b+a6-8 a4+10
(%i12) b^4+4*a*b^3+6*a^2*b^2-2*b^2+4*a^3*b-4*a*b+a^4-2*a^2+1;
(%o12) b4+4 a b3+6 a2 b2-2 b2+4 a3 b-4 a b+a4-2 a2+1
(%i13) factor(%);
(%o13) (b+a-1)2(b+a+1)2
(%i14) expand((a+1)*(b+2)^2-3*(a+b)^6);
(%o14) -3 b6-18 a b5-45 a2 b4-60 a3 b3-45 a4 b2+a b2+b2-18 a5 b+4 a b+4 b-3 a6+4 a+4
```

02-3.- Polinomis i fraccions racionals

Els càlculs d'aquest apartat es desenvolupen al fitxer **Tema_02-3.wxm**.

Sovint els càlculs amb polinomis comporten haver d'aplicar-los més d'una vegada amb un mateix polinomi; això aconsella fer servir una assignació a cada un dels polinomis que s'apliquen als càlculs. La notació vista per a les variables permet aquesta assignació.

Els polinomis són combinacions lineals de les potències d'una variable. Es poden definir amb coeficients expressats com paràmetres generals. Així, per exemple:

```
(%i1) P1:a0+a1*x;  
      P2:b0+b1*x+b2*x^2;  
      P3:c0+c1*x+c2*x^2+c3*x^3;  
(%o1) a1 x + a0  
(%o2) b2 x^2 + b1 x + b0  
(%o3) c3 x^3 + c2 x^2 + c1 x + c0  
(%i4) P1 = P1:a0+a1*x;  
      P2 = P2:b0+b1*x+b2*x^2;  
      P3 = P3:c0+c1*x+c2*x^2+c3*x^3;  
(%o4) a1 x + a0 = a1 x + a0  
(%o5) b2 x^2 + b1 x + b0 = b2 x^2 + b1 x + b0  
(%o6) c3 x^3 + c2 x^2 + c1 x + c0 = c3 x^3 + c2 x^2 + c1 x + c0
```

Observeu que la resposta (sortida) no correspon a l'ordenació de l'entrada. Les operacions suma i producte es poden dur a terme en aquesta forma general. En el cas del producte, wxMaxima l'escriu com a tal sense desenvolupar i si es vol efectuar el producte i expressar-lo com a polinomi en la forma estàndard, cal explicitar-ho amb la instrucció “expand” o bé amb la instrucció “ratsimp” que expressa el polinomi en la forma estàndard.

Així, en el cas dels polinomis anteriors es tindria en el cas de la suma:

```
(%i7) P1 + P2;      P1+P3;  
(%o7) b2 x^2 + b1 x + a1 x + b0 + a0  
(%o8) c3 x^3 + c2 x^2 + c1 x + a1 x + c0 + a0  
(%i9) ratsimp(P1 + P2);      ratsimp(P1 + P3);  
(%o9) b2 x^2 + (b1 + a1) x + b0 + a0  
(%o10) c3 x^3 + c2 x^2 + (c1 + a1) x + c0 + a0
```

En el cas del producte es compleix:


```
(%i11) P1*P2;
(%o11) (a1 x+a0)(b2 x^2+b1 x+b0)
(%i12) P1*P3;
(%o12) (a1 x+a0)(c3 x^3+c2 x^2+c1 x+c0)
(%i13) ratsimp(P1*P2);
(%o13) a1 b2 x^3+(a0 b2+a1 b1) x^2+(a0 b1+a1 b0) x+a0 b0
(%i14) ratsimp(P1*P3);
(%o14) a1 c3 x^4+(a0 c3+a1 c2) x^3+(a0 c2+a1 c1) x^2+(a0 c1+a1 c0) x+a0 c0
```

Observeu que la instrucció “expand” no produeix cap efecte addicional en la suma, que es dona com a resultat en la forma estàndard, mentre que sí afecta al producte i el desenvolupa tal com s’ha indicat.

També es poden definir polinomis amb arrels conegudes expressades en forma de paràmetres, escrits en forma factorial; per exemple:

```
(%i15) P4:C*(x-a1)*(x-a2)^2;
(%o15) (x-a1)(x-a2)^2 C
(%i16) ratsimp(P4);
(%o16) (x^3+(-2 a2-a1) x^2+(a2^2+2 a1 a2) x-a1 a2^2) C
```

A continuació s’il·lustra la sintaxi de la definició i les operacions amb polinomis amb exemples concrets. Les operacions suma i diferència de polinomis es duen a terme de forma senzilla tenint en compte l’assignació realitzada:

```
(%i1) P1:x^2-4;
      P2:x^3-x^2-x+1;
(%o1) x^2-4
      (%o2) x^3-x^2-x+1
(%i3) P1 + P2;      P1 - P2;
(%o3) x^3-x-3
      (%o4) -x^3+2 x^2+x-5
(%i5) P1*P2;
(%o5) (x^2-4)(x^3-x^2-x+1)
(%i6) expand(P1*P2);
(%o6) x^5-x^4-5 x^3+5 x^2+4 x-4
```

Es poden dur a terme combinacions lineals de polinomis, el resultat de les quals s’obté en primera instància en forma simbòlica i si es vol en forma desenvolupada cal fer servir, tal com s’ha indicat, la instrucció “expand”:

```
(%i7) 3*P1-5*P2;
(%o7) 3 (x^2-4)-5 (x^3-x^2-x+1)
(%i8) expand(3*P1-5*P2);
(%o8) -5 x^3+8 x^2+5 x-17
```

Una aplicació útil i important és la descomposició factorial d'un polinomi, que s'obté directament amb la instrucció “factor” (anàloga a l'empira en el cas dels nombres naturals per obtenir la descomposició en factors):

```
(%i9) P3:x^6-1;
(%o9) x^6-1
(%i10) factor(P3);
(%o10) (x-1)(x+1)(x^2-x+1)(x^2+x+1)
```

Es poden calcular amb la instrucció “allroots” les arrels d'un polinomi:

```
(%i11) allroots(P1);    allroots(P2);
(%o11) [x=2.0,x=-2.0]
(%o12) [x=0.99999998509885,x=1.000000014901168,x=-1.000000000000001]
```

En el segon cas, els algorismes numèrics del programa produeixen un resultat que pot semblar sorprenent; la interpretació d'aquest resultat permet concloure que les arrels són 1 (doble) i -1 (simple), aspecte que podem comprovar fàcilment amb la descomposició factorial del polinomi P2:

```
(%i13) factor(P2);
(%o13) (x-1)^2 (x+1)
```

Es pot calcular de forma directa el quocient i del residu de la divisió de dos polinomis amb la instrucció “divide”:

```
(%i14) divide(P2,P1);
      divide(x^4-x^3+x-1, x+1);
      divide(x^4-x^3+x-1, x+2);
(%o14) [x-1,3 x-3]
(%o15) [x^3-2 x^2+2 x-1,0]
(%o16) [x^3-3 x^2+6 x-11,21]
```

També es pot obtenir directament el màxim comú divisor i del mínim comú múltiple de dos polinomis; en aquest segon cas cal carregar prèviament la llibreria “functs”:

```
(%i17) gcd(x^2-2*x+1, x^2-1);
(%o17) x-1
(%i18) load(funcs)$
define: warning: redefining the built-in function lcm
(%i19) lcm(x^2-2*x, x^2-3*x);
(%o19) (x-3)(x-2)x
```

Les fraccions racionals es defineixen de forma senzilla com quocient entre dos polinomis; així mateix, es pot obtenir la descomposició d'una fracció racional en elements o fraccions simples amb la instrucció “partfrac”:

```
(%i1) P1:x^2-4;
      P2:x^3-x^2-x+1;
(%o1) x^2-4
      (%o2) x^3-x^2-x+1
(%i3) P1/P2;
(%o3) 
$$\frac{x^2-4}{x^3-x^2-x+1}$$

(%i4) P1/P2 = partfrac(P1/P2, x);
(%o4) 
$$\frac{x^2-4}{x^3-x^2-x+1} = -\frac{3}{4(x+1)} + \frac{7}{4(x-1)} - \frac{3}{2(x-1)^2}$$

```

Es poden definir primer els polinomis i després la fracció racional amb el quocient dels polinomis considerats:

```
(%i5) Q1:x^2-2*x+4;
      Q2:x^8+5*x^7+13*x^6+22*x^5+26*x^4+22*x^3+13*x^2+5*x+1;
(%o5) x^2-2 x+4
      (%o6) x^8+5 x^7+13 x^6+22 x^5+26 x^4+22 x^3+13 x^2+5 x+1
(%i7) '(Q1/Q2) = partfrac(Q1/Q2, x);
(%o7) 
$$\frac{Q1}{Q2} = \frac{-17x-7}{x^2+x+1} + \frac{-10x-7}{(x^2+x+1)^2} + \frac{-3x-6}{(x^2+x+1)^3} + \frac{17}{x+1} + \frac{7}{(x+1)^2}$$

```

La descomposició d'una fracció racional en fraccions simples es pot dur a terme amb polinomis que continguin paràmetres. Per exemple:

```
(%i8) Q3:x+1;
      Q4:(x-a)*(x^2-b^2);
(%o8) x+1
      (x-a)(x^2-b^2)
(%i10) Q3/Q4 = partfrac(Q3/Q4, x);
(%o10) 
$$\frac{x+1}{(x-a)(x^2-b^2)} = -\frac{b-1}{(2b^2+2ab)(x+b)} + \frac{b+1}{(2b^2-2ab)(x-b)} - \frac{a+1}{(b^2-a^2)(x-a)}$$

```

02-4.- Resolució bàsica d'equacions

Els càlculs d'aquest apartat corresponen al fitxer **Tema_02-4.wxm**.

Les equacions es poden introduir amb la metodologia vista anteriorment de la sintaxi de variables; això permet recuperar-les si cal més endavant i no cal introduir-les novament. Exemples de sintaxi i introducció d'equacions:

```
(%i1) Eq01:2*x+1=x-4;  
(%o1) 2 x + 1 = x - 4  
(%i2) Eq02:a*x+b=c;  
(%o2) a x + b = c  
(%i3) Eq03:4*x^2=5-x;  
(%o3) 4 x^2 = 5 - x  
(%i4) Eq04:x^3-5*x^2-17*x+21=0;  
(%o4) x^3 - 5 x^2 - 17 x + 21 = 0  
(%i5) Eq05:x^4-5*x^3-43*x^2+53*x+90=0;  
(%o5) x^4 - 5 x^3 - 43 x^2 + 53 x + 90 = 0
```

Es poden explicitar els dos membres d'una equació introduïda i definir una nova equació a partir dels membres d'altres equacions:

```
(%i6) first(Eq03);  
      second(Eq03);  
      Eq06:second(Eq03)=first(Eq04);  
(%o6) 4 x^2  
(%o7) 5 - x  
(%o8) 5 - x = x^3 - 5 x^2 - 17 x + 21
```

També es pot definir una nova equació a partir d'operacions amb altres equacions:

```
(%i9) Eq07:Eq03+Eq04;  
(%o9) x^3 - x^2 - 17 x + 21 = 5 - x  
(%i10) Eq08:2*Eq03-4*Eq04;  
(%o10) 8 x^2 - 4 (x^3 - 5 x^2 - 17 x + 21) = 2 (5 - x)  
(%i11) expand(%);  
(%o11) -4 x^3 + 28 x^2 + 68 x - 84 = 10 - 2 x
```

Per a la resolució d'equacions cal fer servir la instrucció “solve” indicant la referència assignada a l'equació i la variable emprada en l'equació:

```
(%i12) solve(Eq01, x);
(%o12) [x = -5]
(%i13) solve(Eq02, x);
(%o13) [x =  $\frac{c-b}{a}$ ]
(%i14) solve(Eq03, x);
(%o14) [x =  $-\frac{5}{4}$ , x = 1]
(%i15) solve(Eq04, x);
(%o15) [x = 1, x = -3, x = 7]
(%i16) solve(Eq05, x);
(%o16) [x = 9, x = -1, x = 2, x = -5]
(%i17) solve(Eq07, x);
(%o17) [x = -4, x = 4, x = 1]
```

Per defecte el programa dona les solucions d'una equació en forma “exacta” o simbòlica, si això és possible. Així, per exemple, les solucions que s’obtenen de les equacions Eq08 i Eq09 són:

```
(%i18) solve(Eq08, x);
(%o18) [x =  $-\frac{\sqrt{130}-6}{2}$ , x =  $\frac{\sqrt{130}+6}{2}$ , x = 1]
(%i19) Eq09:1.44*x^3-7.488*x^2-4.59*x+46.575=0;
(%o19) 1.44 x3 - 7.488 x2 - 4.59 x + 46.575 = 0
(%i20) solve(Eq09, x);
rat: replaced 46.575 by 1863/40 = 46.575
rat: replaced -4.59 by -459/100 = -4.59
rat: replaced -7.488 by -936/125 = -7.488
rat: replaced 1.44 by 36/25 = 1.44
(%o20) [x =  $-\frac{23}{10}$ , x =  $\frac{15}{4}$ ]
```

Per obtenir les solucions en forma numèrica cal aplicar la instrucció “numer”. Per exemple, en el mateix cas de les equacions Eq08 i Eq09 s’obtenen les solucions:

```
(%i21) solve(Eq08, x), numer;
rat: replaced 22.80350850198276 by 6499/285 = 22.80350877192982
rat: replaced 22.80350850198276 by 6499/285 = 22.80350877192982
(%o21) [x = -2.700877192982456, x = 8.700877192982457, x = 1]
```

```
(%i22) solve(Eq09, x), numer;
rat: replaced 46.575 by 1863/40 = 46.575
rat: replaced -4.59 by -459/100 = -4.59
rat: replaced -7.488 by -936/125 = -7.488
rat: replaced 1.44 by 36/25 = 1.44
rat: replaced 46.575 by 1863/40 = 46.575
rat: replaced -4.59 by -459/100 = -4.59
rat: replaced -7.488 by -936/125 = -7.488
rat: replaced 1.44 by 36/25 = 1.44
rat: replaced 0.009 by 9/1000 = 0.009
(%o22) [x = -2.3, x = 3.75]
```

Més endavant es veuran altres mètodes per resoldre equacions, basats en la Teoria de funcions contínues. Així mateix s'estudiaran altres tipus d'equacions no tan senzilles com les estudiades en aquesta Sessió.

02-5.- Resolució bàsica de sistemes d'equacions

Els càlculs d'aquest apartat corresponen al fitxer **Tema_02-5.wxm**.

La sintaxi de definició dels sistemes d'equacions és senzilla i consisteix en assignar una denominació a cada una de les equacions del sistema i finalment una etiqueta al sistema. El programa pot treballar amb sistemes que contenen paràmetres. Per exemple:

```
(%i1) Eq001: a*x+y=a;  
      Eq002:x+a*y=a;  
(%o1) y + a x = a  
(%o2) a y + x = a  
(%i3) Sist01:[Eq001,Eq002];  
(%o3) [y + a x = a, a y + x = a]
```

Es pot organitzar tot la informació en una sola línia d'entrada; per exemple:

```
(%i4) Eq003:x+2*y=0;      Eq004:3*x+5*y=6;  
      Sist02:[Eq003,Eq004];  
(%o4) 2 y + x = 0  
(%o5) 5 y + 3 x = 6  
(%o6) [2 y + x = 0, 5 y + 3 x = 6]
```

Es poden considerar i resoldre sistemes amb un major nombre d'equacions; per exemple si el sistema té tres equacions:

```
(%i7) Eq010:x+y-z=0$      Eq011:x-2*y+z=0$      Eq012:2*x-y+2*z-6=0$  
      Sist03:[Eq010,Eq011,Eq012];  
(%o10) [-z+y+x=0, z-2 y+x=0, 2 z-y+2 x-6=0]
```

La resolució es duu a terme amb la mateixa instrucció o fent servir les opcions del menú Ecuaciones, “Resolver sistema lineal” quan es tracta de sistemes d'equacions lineals:

```
(%i11) solve(Sist01, [x,y]);  
(%o11) [[x =  $\frac{a}{a+1}$ , y =  $\frac{a}{a+1}$ ]]  
(%i12) solve(Sist02, [x,y]);  
(%o12) [[x = 12, y = -6]]  
(%i13) solve(Sist03, [x,y,z]);  
(%o13) [[x = 1, y = 2, z = 3]]
```

Es poden resoldre també sistemes d'equacions no lineals, aplicant la instrucció “algsys” o bé en el menú Ecuaciones, “Resolver sistema algebraico”:


```

(%i14) Eq020:x^4-y^2=1;      Eq021:x^2+y=2;
      Sist04:[Eq020,Eq021];
(%o14)  $x^4 - y^2 = 1$ 
      (%o15)  $y + x^2 = 2$ 
      (%o16)  $[x^4 - y^2 = 1, y + x^2 = 2]$ 
(%i17) algsys([Eq020, Eq021], [x,y]);
(%o17)  $[[x = -\frac{\sqrt{5}}{2}, y = \frac{3}{4}], [x = \frac{\sqrt{5}}{2}, y = \frac{3}{4}]]$ 
(%i18) Eq022:3*x^2-y^2=6;      Eq023:x-y=0;
      Sist05:[Eq022,Eq023];
(%o18)  $3x^2 - y^2 = 6$ 
      (%o19)  $x - y = 0$ 
      (%o20)  $[3x^2 - y^2 = 6, x - y = 0]$ 
(%i21) algsys([Eq022, Eq023], [x,y]);
(%o21)  $[[x = -\sqrt{3}, y = -\sqrt{3}], [x = \sqrt{3}, y = \sqrt{3}]]$ 

```

02-6.- Resolució bàsica d'inequacions

Els càlculs d'aquest apartat corresponen al fitxer **Tema_02-6.wxm**.

Les inequacions tenen una sintaxi senzilla, similar a la de les equacions. Suposem, per exemple, que es vol resoldre la inequació $\frac{1}{x-2} < 1$; la sintaxi en wxMaxima és:

```
(%i1) Ineq01:1/(x-2)<1;  
(%o1)  $\frac{1}{x-2} < 1$ 
```

Ara observem que la inequació es pot transformar de la manera següent:

$$\frac{1}{x-2} < 1 \Leftrightarrow \frac{1}{x-2} - 1 < 0 \Leftrightarrow \frac{1-(x-2)}{x-2} = \frac{3-x}{x-2} < 0$$

Per tant, s'escriurà:

```
(%i2) Ineq01b:(3-x)/(x-2)<0;  
(%o2)  $\frac{3-x}{x-2} < 0$ 
```

La resolució d'una inequació es duu a terme amb la introducció de una condició o condicions sobre el domini de la variable i la verificació del signe de la condició corresponent a la inequació; així en el cas anterior cal considerar els casos $x < 2$, $2 < x < 3$ i $x > 3$.

La sintaxi és la següent:

```
(%i3) assume(x<2);  
      is((3-x)/(x-2)<0);  
(%o3) [x < 2]  
(%o4) true
```

Així doncs, la inequació és certa en aquest cas; ara cal introduir la nova condició però també cal eliminar la condició introduïda anteriorment:

```
(%i5) forget(x<2)$  
      assume(x>2, x<3);  
      is((3-x)/(x-2)<0);  
(%o6) [x > 2, x < 3]  
(%o7) false
```

Continuant amb la metodologia, eliminarem la darrera condició introduïda i en posarem una de nova, per acabar de resoldre la inequació:

```
(%i8) forget(x>2, x<3)$  
      assume(x>3);  
      is((3-x)/(x-2)<0);  
(%o9) [ x > 3 ]  
(%o10) true
```

Finalment, doncs, es pot concloure que la inequació és certa en l'interval $]-\infty, 2[\cup]3, +\infty[$.

02-7.- Funcions elementals amb wxMaxima

Els càlculs d'aquest apartat corresponen al fitxer **Tema_02-7.wxm**.

A la Taula 02-1 es pot veure la descripció de les funcions elementals i la seva notació o sintaxi en wxMaxima. És convenient saber aquesta notació per tal d'aplicar correctament els càlculs que involucrin les funcions esmentades.

Funció	Notació matemàtica usual	Notació wxMaxima
Valor absolut de x	$ x $	<code>abs(x)</code>
Arrel quadrada de x	$\sqrt{x}$	<code>sqrt(x)</code>
Exponencial real	$\exp(x)$	<code>exp(x)</code>
Logaritme neperià	$\ln(x)$ o bé $\log(x)$	<code>log(x)</code>
Sinus	$\sin(x)$	<code>sin(x)</code>
Cosinus	$\cos(x)$	<code>cos(x)</code>
Tangent	$\tan(x)$	<code>tan(x)</code>
Arc sinus	$\arcsin(x)$	<code>asin(x)</code>
Arc cosinus	$\arccos(x)$	<code>acos(x)</code>
Arc tangent	$\arctan(x)$	<code>atan(x)</code>
Sinus hiperbòlic	$\sinh(x)$	<code>sinh(x)</code>
Cosinus hiperbòlic	$\cosh(x)$	<code>cosh(x)</code>
Tangent hiperbòlica	$\tanh(x)$	<code>tanh(x)</code>
Argument sinus hiperbòlic	$\operatorname{asinh}(x)$	<code>asinh(x)</code>
Argument cosinus hiperbòlic	$\operatorname{acosh}(x)$	<code>acosh(x)</code>
Argument tangent hiperbòlica	$\operatorname{atanh}(x)$	<code>atanh(x)</code>

Taula 02-1 Funcions elementals predefinides en wxMaxima: denominació, notació habitual i notació de wxMaxima.

Per calcular el valor d'una funció predefinida en un punt només cal escriure la notació de la funció i el punt a continuació entre parèntesi. Així, per exemple, la sintaxi per treballar amb les funcions “valor absolut”, “arrel quadrada” i “exponencial” és:

```
(%i1) abs((-2.35)^3-4*(7.25^2));
(%o1) 223.227875

(%i2) sqrt(8975.3456); sqrt(10); sqrt(10.0);
(%o2) 94.73830059696026

(%o3)  $\sqrt{10}$ 

(%o4) 3.16227766016838
```

```
(%i5) exp(0);    exp(1);    exp(1.0);    exp(10.45);
(%o5) 1
(%o6) %e
(%o7) 2.718281828459045
(%o8) 34544.37470927775
```

S'observa que l'arrel quadrada d'un nombre natural es dona indicant el nombre sota el símbol de l'arrel quadrada; si es vol el resultat en forma numèrica cal fer servir la instrucció “numer” o introduir el nombre amb coma flotant per tal que sigui interpretat com a nombre real.

Vegem ara el càlcul del valor de la funció logaritme neperià (o logaritme natural) en diferents punts; cal fer notar que en wxMaxima la funció logaritme neperià es designa “log”, seguint la tendència més moderna en Anàlisi Matemàtica i que alguns textos ja van incorporant; la notació clàssica, que encara és la més habitual, és “ln”. Alguns exemples de càlculs:

```
(%i9) log(0.5);    log(1.0);    log(2.0);    log(%e);
      log(%e^2);    log(%e^n);    log(10);    log(10.0);
(%o9) -0.69314718055995
(%o10) 0.0
(%o11) 0.69314718055995
(%o12) 1
(%o13) 2
(%o14) n
(%o15) log(10)
(%o16) 2.302585092994046
```

Una qüestió important és el canvi de base en la funció logarítmica, que permet la definició de la funció logarítmica de base qualsevol nombre $b > 0, b \neq 1$:

```
(%i17) log_b(x):=(log(x)/log(b));    b:12.0$    x:2.0$    log_b(x);
(%o17) log_b(x):= $\frac{\log(x)}{\log(b)}$ 
(%o20) 0.27894294565113
(%i21) b:12.3450$    x:2.98760$    log_b(x);
(%o23) 0.43547991665977
```

L'argument de les funcions trigonomètriques per defecte és en radians; per calcular el valor de les funcions trigonomètriques d'un angle expressat en graus sexagesimals o centesimals, cal efectuar la conversió del corresponent valor en radians:

$$\text{Angle (radians)} = \text{Angle (graus sexagesimals)} \cdot \frac{\pi}{180}$$

$$\text{Angle (radians)} = \text{Angle (graus centesimals)} \cdot \frac{\pi}{200}$$

Càlcul del sinus de diversos angles:

```
(%i1) x1:1.0$    x2:%pi$    x3:3*(%pi)/2$    x4:2*(%pi)$
      sin(x1);    sin(x2);    sin(x3);    sin(x4);
(%o5) 0.8414709848079
(%o6) 0
(%o7) -1
(%o8) 0
```

Càlcul del cosinus de diversos angles:

```
(%i9) cos(x1); cos(x2); cos(x3); cos(x4);
(%o9) 0.54030230586814
(%o10) -1
(%o11) 0
(%o12) 1
```

Definició de les funcions sinus i el cosinus quan l'angle s'expressa en graus sexagesimals:

```
(%i13) sin_g(x):=sin(x*(%pi)/180);
      cos_g(x):=cos(x*(%pi)/180);
(%o13) sin_g(x):=sin\left(\frac{x \pi}{180}\right)
(%o14) cos_g(x):=cos\left(\frac{x \pi}{180}\right)
```

Exemples de càlculs amb aquestes funcions:

```
(%i15) y1:30$
      y2:45$
      y3:60$
      y4:90$
```

```
(%i19) sin_g(y1); sin_g(y2); sin_g(y3); sin_g(y4);
(%o19)  $\frac{1}{2}$ 
(%o20)  $\frac{1}{\sqrt{2}}$ 
(%o21)  $\frac{\sqrt{3}}{2}$ 
(%o22) 1
(%i23) cos_g(y1); cos_g(y2); cos_g(y3); cos_g(y4);
(%o23)  $\frac{\sqrt{3}}{2}$ 
(%o24)  $\frac{1}{\sqrt{2}}$ 
(%o25)  $\frac{1}{2}$ 
(%o26) 0
```

Definició del sinus i el cosinus quan l'angle s'expressa en graus centesimals:

```
(%i27) sin_c(x):=sin(x*(%pi)/200);
      cos_c(x):=cos(x*(%pi)/200);
(%o27) sin_c(x):=sin( $\frac{x \pi}{200}$ )
(%o28) cos_c(x):=cos( $\frac{x \pi}{200}$ )
```

Exemples de càlculs amb aquestes funcions:

```
(%i29) y1:25$ y2:50$ y3:100$
      sin_c(y1), numer; sin_c(y2), numer; sin_c(y3), numer;
(%o32) 0.38268343236509
(%o33) 0.70710678118655
(%o34) 1.0
(%i35) y1:25$ y2:50$ y3:100$
      cos_c(y1),numer; cos_c(y2),numer; cos_c(y3),numer;
(%o38) 0.92387953251129
(%o39) 0.70710678118655
(%o40) 6.1230317691118863 10-17
```

Càlcul del valor de la funció tangent en diversos punts:

```
(%i41) tan(%pi/4); tan(%pi); tan(3*%pi/4); tan(5*%pi/14);
(%o41) 1
(%o42) 0
(%o43) -1
(%o44)  $\tan\left(\frac{5\pi}{14}\right)$ 
(%i45) tan(5*%pi/14), numer;
(%o45) 2.076521396572336
```

Càlcul del valor de la funció arc-sinus en diversos punts:

```
(%i46) asin(0); asin(0.5); asin(1);
(%o46) 0
(%o47) 0.5235987755983
(%o48)  $\frac{\pi}{2}$ 
```

Càlcul del valor de la funció arc-cosinus en diversos punts:

```
(%i49) acos(0); acos(0.5); acos(1);
(%o49)  $\frac{\pi}{2}$ 
(%o50) 1.047197551196598
(%o51) 0
```

Càlcul del valor de la funció arc-tangent en diversos punts:

```
(%i52) atan(0); atan(2.0); atan(5.0);
(%o52) 0
(%o53) 1.10714871779409
(%o54) 1.373400766945016
```

Càlcul del valor de la funció sinus hiperbòlic en diversos punts:

```
(%i1) sinh(0); sinh(2.0); sinh(5.0);
(%o1) 0
(%o2) 3.626860407847019
(%o3) 74.20321057778875
```

Càlcul del valor de la funció cosinus hiperbòlic en diversos punts:


```
(%i4) cosh(0);      cosh(2.0);      cosh(5.0);
(%o4) 1
(%o5) 3.762195691083631
(%o6) 74.20994852478785
```

Càlcul del valor de la funció tangent hiperbòlica en diversos punts:

```
(%i7) tanh(0);      tanh(2.0);      tanh(5.0);
(%o7) 0
(%o8) 0.96402758007582
(%o9) 0.9999092042626
```

La sintaxi per a la definició d'una funció ja s'ha comentat a l'apartat anterior. L'estudi més exhaustiu de les funcions es veurà en sessions posteriors. A continuació es mostra un exemple de funció definida mitjançant operacions amb funcions elementals i el càlcul del seu valor en tres punts del seu camp d'existència.

```
(%i1) f(x):=2*sin(x)-3*exp(x)+log(abs((x-1)/(x-2)));
(%o1) f(x):= 2 sin(x)-3 exp(x)+log( $\left(\frac{x-1}{x-2}\right)$ )
(%i2) f(3.45); f(-2.54); f(2.0001);
(%o2) -94.58373548353652
(%o3) -1.617311945177004
(%o4) -11.14043314229555
```